Information Circular 8871

A Computer Program for Calculating Thermodynamic Properties From Spectroscopic Data

By R. P. Beyer

UNITED STATES DEPARTMENT OF THE INTERIOR
James G. Watt, Secretary

BUREAU OF MINES
Robert C. Horton, Director

/ a



This publication has been cataloged as follows:

Beyer, R. P. (Richard P.)

A computer program for calculating thermodynamic properties from spectroscopic data.

(Information circular ; 8871)

Bibliography: p. 19.

Supt. of Docs. no.: I 28.23:8871.

1. Thermodynamics—Computer programs. 2. Gases—Thermal properties—Computer programs. 3. Gases—Spectra—Computer programs. 4. FORTRAN (Computer program language). I. Title. II. Series: Information circular (United States. Bureau of Mines) ; 8871.

TN295.U4 [TJ265] 622s [536'.25] 81-607926 AACR2

CONTENTS

SYMBOLS AND TERMINOLOGY[1]

e_i = energy of the ith electronic level, cm^{-1}
g_i = degeneracy of the i_{th} electronic level
σ = symmetry number
α = rotational vibrational interaction constant
x_e = anharmonicity correction
ω_e = fundamental frequency, cm^{-1}
I = moment of interia, g cm^2
S = entropy, cal mol^{-1} K^{-1}
C_p = heat capacity, cal mol^{-1} K^{-1}
H = enthalpy, cal $mol^{-1}K^{-1}$
r_e = bond distance, Å
T_p = transition point temperature, K
h = Planck's constant, $6.62617636 \times 10^{-27}$ erg-sec
k = Boltzmann's constant, $1.38066244 \times 10^{-16}$ erg $molecule^{-1}$ K^{-1}
R = gas constant, 1.98719.7 cal $K^{-1}mol^{-1}$
R^1 = gas constant, 82.056826 atm cm^3 mol^{-1} K^{-1}
N_A = Avogadro's constant, $6.02204531 \times 10^{23}$ mol^{-1}
c = Speed of light, $2.9979245812 \times 10^{10}$ m sec^{-1}
M = molecular weight, g mol^{-1}
ε_i = energy in ergs, (cm^{-1} x hc)

---

[1] Values for the physical constant are from (2). Underlined numbers in parentheses refer to items in the list of references at the end of this paper.

A COMPUTER PROGRAM FOR CALCULATING THERMODYNAMIC PROPERTIES FROM SPECTROSCOPIC DATA

by

R. P. Beyer[2]

ABSTRACT

A FORTRAN IV computer program has been written to calculate thermodynamic properties over a desired temperature range from spectroscopic data. This program computes heat capacities, enthalpies, and entropies for monatomic, diatomic, linear, and nonlinear polyatomic gases. This research is part of the Bureau of Mines effort to provide thermodynamic data for the advancement of mineral technology.

[2]Chemical engineer, Albany Research Center, Bureau of Mines, Albany, Oreg.

INTRODUCTION

As part of a program to advance technology, reduce energy use, and abate pollution, the Bureau of Mines generates thermochemical data on various minerals and metals. A computer program for calculating C_p°, $S^\circ(T)-S^\circ(0)$, and $H^\circ(T)-H^\circ(298)$ was written to further this aim. The program was written in FORTRAN IV Plus, Version 2.5, to run on a Digital Equipment Corp.[3] PDP 11/34 minicomputer with the RSX-11M, Version 3.2, operating system. Although programs have been written previously for this type of calculation, [e.g., McBride and Gordon (5)], the vast differences in hardware, operating systems, and compilers necessitated a rewriting rather than an adaptation of previous work.

The program accepts as input the electronic energy levels, degeneracies, moment of inertia, number of atoms, vibrational frequencies, vibrational-rotational interaction constant, symmetry number, and molecular weight. The output consists of the heat capacities, entropies, and relative enthalpies over a given temperature range. The normal output consists of the thermodynamic properties calculated over the desired temperature range at intervals starting at 0 K and including 298.15 K. Any noninternal desired temperature may also be included.

CALCULATIONAL METHODS

The methods to calculate the thermodynamic properties follow the adaptation of Mayer and Mayer (4) by King and others (3). For the definition of the symbols used, refer to the symbols and terminology section.

(1) Monatomic gases

Translational contribution:

$$(C_p)_{tr} = 5/2\ R$$

$$(H - H_0)_{tr} = 5/2\ RT$$

$$S_{tr} = R(\ell nR^1 + 5/2\ \ell nT + 3/2\ \ell nM + 3/2\ \ell n\frac{2\pi k}{h^2} - 5/2\ \ell nN + 5/2)$$

---

[3]Reference to specific brand names does not imply endorsement by the Bureau of Mines.

Electronic contribution:

$$C_e = R\left[\frac{Q^{II}}{Q} - \left(\frac{Q^I}{Q}\right)^2\right]$$

$$(H - H_0)_e = RT\left[\frac{Q^I}{Q}\right]$$

$$S_e = R\left[\ell nQ + \frac{Q^I}{Q}\right]$$

where

$$Q = \sum_i g_i e^{-\epsilon_i/kT}$$

$$Q^I = \sum_i g_i \left[\frac{\epsilon_i}{kT}\right] e^{-\epsilon_i/kT}$$

$$Q^{II} = \sum_i g_i \left[\frac{\epsilon_i}{kT}\right]^2 e^{-\epsilon_i/kT}$$

(2) Diatomic gases

Translational and electronic contributions: same as for monatomic gases.
Rotational contribution

$$C_r = R\left(1 + \frac{y^2}{45}\right)$$

$$(H - H_0)_r = RT\left(1 - \frac{y}{3} - \frac{y^2}{45}\right)$$

$$S_r = R\left(1 - \ln y - \ln\sigma - \frac{y^2}{90}\right)$$

where $y = \dfrac{hcB_0}{kT}$

$$B_0 = B_e - 1/2 \cdot \alpha cm^{-1}$$

$$B_e = \frac{h}{8\pi^2 cI}\ cm^{-1}$$

Vibrational contribution

$$C_{vib} = R \frac{u^2 e^u}{(e^u - 1)^2}$$

$$(H - H_0)_{vib} = RT \frac{u}{(e^u - 1)}$$

$$S_{vib} = R\left[\frac{u}{e^u - 1} - \ell n\ (1 - e^{-u})\right]$$

where $u = \frac{hc\omega_0}{kT}$

$$\omega_0 = \omega_e - 2x_e\omega_e,\ cm^{-1}$$

Anharmonicity corrections

$$C_{cor} = \frac{R}{u}\left[4x\left[\frac{u^3 e^u (2ue^u - 2e^u + u + 2)}{(e^u - 1)^4}\right] + \delta\left[\frac{u^3 e^u (e^u + 1)}{(e^u - 1)^3}\right] + 16\gamma\right]$$

$$(H - H_0)_{cor} = \frac{RT}{u}\left[2x\left[\frac{u^2\ (2ue^u - e^u + 1)}{(e^u - 1)^3}\right] + \delta\left[\frac{u^2 e^u}{(e^u - 1)^2}\right] + 8\gamma\right]$$

$$S_{cor} = R\left[\frac{4xu^2 e^u}{(e^u - 1)^3} + \frac{\delta u e^u}{(e^u - 1)^2} + \frac{\delta}{(e^u - 1)} + \frac{16\gamma}{u}\right]$$

where $x = \frac{x_e}{1 - 2x_e}$

$$\delta = \frac{\alpha}{B_0}$$

$$\gamma = \frac{B_e}{\omega_e}$$

(3) Linear polyatomic gases

Electronic: Same for all atoms and molecules.
Translational: Same for all atoms and molecules.
Rotational plus symmetry: Same as for diatomic molecules except for calculation of I(moment of inertia).
Vibrational: Fundamental frequencies 3n-5 in number. Follow the diatomic method and sum the results.

(4) Nonlinear polyatomic

Electronic: Same as linear polyatomic.
Translational: Same as linear polyatomic.
Rotational plus symmetry:

$$\left(H(T) - H(0)\right)_r = (3/2)RT.$$

$$C_{p_r} = (3/2)R$$

$$S_r = 3/2R + R\left[1/2\ \ell n\ I_A I_B I_C + 3/2\ \ell nT - \ell n\sigma + 1/2\ \ell n\ \frac{512\pi^7 k^3}{h^6}\right]$$

Vibrational: Fundamental frequencies 3n-6 in number. Follow the diatomic method and sum the results.

PROGRAM OPERATION

Input consists of a data file containing the pertinent information. There are four different types of data input files, depending on which type of gas is being studied. All data in the examples are from JANAF (1). The following examples show the required format.

(1) The monatomic input file:

a) In general,
Name
Monatomic
Highest temperature to be calculated
K, the number of noninterval temperatures (Tp_i, i = 1,2,....K)
•
•
•
Tp_i
•
•
•

```
.
.
.
Number of energy levels
.                                          .
.                                          .
.                                          .
.                                          .
Electronic energy of level i (e_i)         Degeneracy of
.                                           level i (g_i)
.                                          .
.                                          .
.                                          .
.                                          .
Molecular weight
```

b) Monatomic example, F(g):

```
F(g)
Monatomic
2000.
0
12
0                4
404.             2
103327.14       18
115918.7         6
116597.23       14
117465.88       22
118627.73       10
123118.7        12
128346.36       80
132786.07       16
133531.81       22
134978.71       88
18.9984
```

(2) The Diatomic input file:

a) In general,

```
Name
Diatomic
Highest temperature to be calculated
Number of noninterval temperatures
.
.
```

```
.
.
Tp_i
.
.
.
Number of energy levels
.                                                    .
.                                                    .
.                                                    .
.                                                    .
Electronic energy of level i (e_i)        Degeneracy of level i (g_i)
.                                                    .
.                                                    .
.                                                    .
.                                                    .
.                                                    .
.                                                    .
.                                                    .
.                                                    .
Atomic weight of atom 1
Atomic weight of atom 2
Internuclear distance (r_e in Å)
Fundamental frequency (ω_e in cm^-1)
Anharmonicity correction (ω_e x_e)
Rotational-vibrational interaction constant (α)
Symmetry number (σ)
```

b) Diatomic example, CuF(g);

```
CuF(g)
Diatomic
6000
0
1
0           1
63.54
18.9984
1.743
621.89
3.941
0.004586
1
```

(3) The linear polyatomic input file:

a) In general,

Name
Linear polyatomic
Highest temperature to be calculated
Number of noninterval temperatures, (Tp)
•
•
•
•
Tp_i
•
•
•
Number of energy levels
•
•
•
•
Electronic energy of level i (e_i)
•
•
•
•
•
•
Molecular weight
Number of atoms
•
•
•
•
Fundamental frequency of level i (ω_i), enter degenerate frequencies as many times as the degeneracy of the modes, see examples below.
Moment of inertia (I)
Rotational vibrational interaction constant (α)
Symmetry number (σ)

•
•
•
•
Degeneracy of level i (g_i)
•
•
•
•
•

b) Linear polyatomic example, $CuF_2(g)$:

```
CuF2(g)
Linear Polyatomic
2000
0
3
0           2
9000        4
18000       4
101.5368
3
608
205
205
768
112.4008735 (see below for units)
0
2
```

(4) The nonlinear polyatomic input file:

a) In general,

The same as the linear case except the second line should read nonlinear polyatomic and I is the product of the three principal moments of inertia $I = I_A I_B I_C$.

b) The nonlinear polyatomic example $ZrI_4(g)$:

```
ZrI4(g)
Nonlinear polyatomic
2000.
2
155.3
623.9
1
0           1
598.66
5
146.
 45.
 45.
237.
237.
237.
 58.
 58.
 58.
65718000 (see below for units)
0
12
```

The FORTRAN-IV Plus compiler used with this program limits calculations to numbers between $\pm 10^{38}$ and $\pm 10^{-38}$. Therefore, values for the moment of inertia for linear and nonlinear polyatomics should be entered as follows:

(1) Linear: values of I in units of g-cm^2 should be multiplied by $(N_A/10^{-16})$ before entering in the data input file. CuF_2(g), I = 18.7 x 10^{-39} g-cm^2. For the input file this number would be changed to $(18.7 \times 10^{-39})(N_A/10^{-16}) = 112.4$.

(2) Polyatomic: Values for $I_A I_B I_C$, which would have an exponent of 10^{-117} [from $(10^{-39})^3$], should be entered without this exponent. For example, ZrI_4(g), $I_A I_B I_C$ = 65718000 x 10^{-117}. This should be entered as just 65718000.

An example of the output file for the ZrI_4(g) example is given after the program listing. The program was checked by running all the examples shown and comparing the output with the accepted JANAF (1) results. Agreement between the computer output and the JANAF values was exact.

PROGRAM LISTING

```
C
C A PROGRAM TO CALCULATE HEAT CAPACITY, ENTHALPY, AND
C ENTROPY FROM SPECTROSCOPIC DATA
C
      Real*8 HT,e(100),g(100),w(100),xe(100),wexe(100),MW,MW1,MW2
      Real*8 re,a,sig,Be,B0,y,x,c,R,pi,k,h,kT,1ny
      Real*8 ctrans,htrans,strans,crot,hrot,srot,cvib,hvib,svib
      Real*8 celec,helec,selec,cr298,hr298,sr298,ht298,ct298,st298
      Real*8 cv298,hv298,sv298,ce298,he298,se298
      Real*8 q0,q1,q2,MI,Rprime,Na,hout
      Real*8 del,hcor,scor,eu,htotal,h298,stotal,s298,eul,T(100),gam
      Real*8 Td,ccor,ctotal,c298,u,w0(100)
      Byte igas,gasnam(30)
c
c Initialize
c
      do 500 i=1,100
      e(i)=0.0
      T(i)=0.0
      g(i)=0.0
      w(i)=0.0
      w0(i)=0.0
      xe(i)=0.0
  500 wexe(i)=0.0
      Na=6.02204531e23
      Rprime=82.056826
      k=1.38066244e-16
      h=6.62617636e-27
      c=2.9979245812e10
      R=1.9871917
      pi=3.141592654
c
c conversational input of constants
c
      open(unit=1,name='dl1:gas.dat',type='new',dispose='save')
      write(5,130)
  130 format(/,1x,'Enter name of the input file')
      read(5,131)(gasnam(i),i=5,30)
  131 format(30A1)
      type *,'Output is in file gas.dat'
      gasnam(30)=0
      gasnam(1)='S'
      gasnam(2)='Y'
      gasnam(3)='0'
      gasnam(4)=':'
      open(unit=2,name=gasnam,type='old',dispose='save')
      read(2,131)(gasnam(i),i=1,20)
      write(1,132)(gasnam(i),i=1,20)
  132 format(1x,20A1,/)
```

```
      read(2,101)igas
  101 format(1A1)
      if(igas.eq.'M'.or.igas.eq.'m')go to 2
      if(igas.eq.'D'.or.igas.eq.'d')go to 3
      if(igas.eq.'L'.or.igas.eq.'l')go to 4
      if(igas.eq.'N'.or.igas.eq.'n')go to 5
      type *,'INVALID GAS TYPE!'
      stop
    2 type *,' monatomic gas'
      iflag=7
      go to 6
    3 type *,' diatomic gas'
      iflag=8
      go to 6
    4 type *,' linear polyatomic gas'
      iflag=9
      go to 6
    5 type *,' non-linear polyatomic gas'
      iflag=9
      go to 6
    6 read(2,*)HT
      iHT=HT
      ntemp=iHT/100
      T(1)=298.15
      do 43 i=1,ntemp
   43 T(i+1)=100.*i
c
c Read tranisition temperatures
c
      read(2,*)nTp
      if(nTp.eq.0)go to 41
      do 42 i=1,nTp
   42 read(2,*)T(ntemp+1+i)
c
c Arrange the temperatures in ascending order
c
      do 44 j=2,ntemp+nTp
      do 44 i=2,ntemp+nTp
      if(T(i).lt.T(i+1))go to 44
      Ttemp=T(i)
      T(i)=T(i+1)
      T(i+1)=Ttemp
   44 continue
   41 read(2,*)nlevel
      do 50 i=1,nlevel
   50 read(2,*)e(i),g(i)
      if(igas.eq.'M'.or.igas.eq.'m')go to 7
      if(igas.eq.'D'.or.igas.eq.'d')go to 8
      go to 9
c
c monatomic gas
c
    7 read(2,*)MW
      go to 20
```

```
c
c diatomic gas
c
    8 read(2,*)MW1
      read(2,*)MW2
      read(2,*)re
c
c calculate moment of inertia
c
      MI=((MW1*MW2)/(MW1+MW2))*re*re
      MW=MW1+MW2
      read(2,*)w(1)
      read(2,*)wexe(1)
      read(2,*)a
      read(2,*)sig
      nvib=1
      go to 20
c
c polyatomic gas
c
    9 read(2,*)MW
      read(2,*)natom
      nvib=3*natom-6
      if(igas.eq.'L'.or.igas.eq.'l')nvib=3*natom-5
      do 51 i=1,nvib
   51 read(2,*)w(i)
      read(2,*)MI
      read(2,*)a
      read(2,*)sig
      iharm=0
c
   20 write(5,300)
  300 format(//,6x,'T',10x,'Cp',10x,'H-H298',11x,'S')
      write(1,300)
      write(1,302)
  302 format(6x,'K',7x,'cal/mol-K',6x,'kcal/mol',5x,'cal/mol-K',/)
c
c  start calculations
c
      ht298=0.0
      hr298=0.0
      he298=0.0
      hv298=0.0
      hc298=0.0
      istart=1
   17 do 55 i=istart,ntemp+nTp+1
      if(igas.eq.'M'.or.igas.eq.'m')go to 16
c
c rotational calulations
c
   12 if(igas.eq.'D'.or.igas.eq.'d')go to 13
      if(igas.eq.'L'.or.igas.eq.'l')go to 13
c
c non-linear polyatomic rotational
```

```
14
c
      crot=1.5*R
      hrot=T(i)*crot+hr298
      srot=(R/2.)*(log(512.*pi**7*((k*1.e16)**3)/((h*1.e27)**6))
     &-3.*log(10.)+3.)+(R/2.)*(log(MI)+3.*log(T(i))-2.*log(sig))
      go to 14
c
c diatomic or linear polyatomic rotational
c
   13 Be=((h*Na)/(8.*pi*pi*c*1.e-16))/MI
      B0=Be-0.5*a
      y=(((h*Na)*h/(k*8.*pi*pi*1.e-16))/(T(i)*MI))
     &-((h*c)/(2.*k))*(a/T(i))
      crot=R*(1.+(y*y/45.))
      hrot=R*T(i)*(1.-y/3.-(y*y/45.))+hr298
      lny=0.0
      if(y.gt.0.0)lny=log(y)
      srot=R*(1.-lny-log(sig)-(y*y/90.))
c
c vibrational calculations
c
   14 cvib=0.0
      hvib=0.0
      svib=0.0
      ccor=0.0
      hcor=0.0
      scor=0.0
      do 53 j=1,nvib
      w0(j)=w(j)-2.*wexe(j)
      u=((h*c)/k)*w0(j)/T(i)
      eu=exp(u)
      cvib=(R*u*u*eu/((eu-1.)*(eu-1.)))+cvib
      hvib=(R*T(i)*u/(eu-1.))+hvib
      svib=(R*((u/(eu-1.))-log(1.-exp(-u))))+svib
c
c anharmonicity corrections
c
      if(nvib.eq.1)go to 15
      if(iharm.ne.1)go to 53
   15 xe(j)=wexe(j)/w(j)
      del=0.0
      if(a.ne.0.0)del=a/(Be-a/2.)
      gam=Be/w(j)
      eul=eu-1.
      x=xe(j)/(1.-2.*xe(j))
      ccor=(R/u)*((4.*x*u*u*u*eu*(2.*u*eu-2.*eu+u+2.)/(eul*eul*eul*eul))
     &+(del*u*u*u*eu*(eu+1.)/(eul*eul*eul))+16.*gam)+ccor
      hcor=(R*T(i)/u)*(((2.*x)*u*u*(2.*u*eu-eu+1.)/(eul*eul*eul))+
     &(del*u*u*eu/(eul*eul))+8.*gam)+hcor
      scor=R*((4.*x*u*u*eu/(eul*eul*eul))+(del*u*eu/(eul*eul))+(del/eul)
     &+(16.*gam/u))+scor
   53 continue
      hvib=hvib+hv298
      hcor=hcor+hc298
```

```
c
c Translational calculations
c
   16 ctrans=2.5*R
      htrans=ctrans*T(i)+ht298
      strans=R*(2.5*log(T(i))+1.5*log(MW)
     &+log(Rprime)+1.5*log(2.*pi*k)-3*log(h)-2.5*log(Na)+2.5)
c
c electronic calculations
c
      celec=0.0
      helec=0.0
      selec=0.0
      call Q(e,g,nlevel,T(i),q0,q1,q2)
      if(q0.eq.0)go to 22
      celec=R*((q2/q0)-(q1/q0)*(q1/q0))
      helec=R*T(i)*q1/q0+he298
      selec=R*(log(q0)+(q1/q0))
c
c calculate total thermodynamic functions
c
   22 ctotal=ctrans+celec+crot+cvib+ccor
      htotal=htrans+hrot+hvib+helec+hcor
      stotal=strans+srot+svib+selec+scor
c
c if T=298.15 then go print values
c
      if(i.eq.1)go to 18
      jflag=0
      if(T(i).eq.200)jflag=1
      write(5,122)T(i),ctrans,htrans,strans
  122 format(/,1x,F9.2,3(1x,f12.4))
      write(5,123)celec,helec,selec
  123 format(10x,3(1x,f12.4))
      if(igas.eq.'M'.or.igas.eq.'m')go to 21
      write(5,123)crot,hrot,srot
      write(5,123)cvib,hvib,svib
      write(5,123)ccor,hcor,scor
   21 write(5,123)ctotal,htotal,stotal
      hout=htotal/1000.
      write(1,301)T(i),ctotal,hout,stotal
  301 format(1x,f8.2,3(3x,f11.3))
      if(jflag.eq.0)go to 55
      T(i)=298.15
      ccor=cc298
      hcor=hc298
      scor=sc298
      crot=cr298
      hrot=0.0
      srot=sr298
      ctrans=ct298
      htrans=0.0
      strans=st298
      cvib=cv298
```

```
      hvib=0.0
      svib=sv298
      celec=ce298
      helec=0.0
      selec=se298
      hcor=0.0
      go to 22
   55 continue
      go to 19
   18 ht298=htrans
      hr298=hrot
      he298=helec
      hv298=hvib
      hc298=hcor
      h298=htotal
      cr298=crot
      ct298=ctrans
      cv298=cvib
      cc298=ccor
      ce298=celec
      sr298=srot
      st298=strans
      sv298=svib
      sc298=scor
      se298=selec
      c298=ctotal
      s298=stotal
      write(5,124)ht298
  124 format(5x,'0',13x,'0',4x,f12.4,10x,'0',5x,'Translational')
      write(5,125)he298
  125 format(19x,'0',4x,f12.4,10x,'0',5x,'Electronic')
      if(igas.eq.'M'.or.igas.eq.'m')go to 25
      write(5,200)hr298
  200 format(19x,'0',4x,f12.4,10x,'0',5x,'Rotational')
      write(5,201)hv298
  201 format(19x,'0',4x,f12.4,10x,'0',5x,'Vibrational')
      write(5,202)hc298
  202 format(19x,'0',4x,f12.4,10x,'0',5x,'Anharmonic')
   25 write(5,203)h298
  203 format(19x,'0',4x,f12.4,10x,'0',5x,'Total')
      hout=h298/1000.
      write(1,204)hout
  204 format(8x,'0',13x,'0',3x,f11.3,13x,'0')
      istart=2
      go to 17
   19 continue
      close (unit=1)
      end
c
c subroutine to calculate partition functions
c
      subroutine Q(e,g,nlevel,T,q0,q1,q2)
      Real*8 e(nlevel),g(nlevel),T,q0,q1,q2,kT,ekT,ge
      kT=1.438785935/T
```

```
      q0=0.0
      q1=0.0
      q2=0.0
      do 1 i=1,nlevel
      ekT=e(i)*kT
      ge=exp(-ekT)*g(i)
      q0=q0+ge
      q1=q1+ekT*ge
    1 q2=q2+ekT*ekT*ge
      return
      end
```

PROGRAM OUTPUT

ZrI4(g)

T	Cp	H-H298	S
K	cal/mol-K	kcal/mol	cal/mol-K
0	0	-6.330	0
100.00	21.258	-4.743	80.982
200.00	24.284	-2.430	96.897
298.15	25.090	0.000	106.772
300.00	25.098	0.046	106.927
400.00	25.410	2.574	114.196
500.00	25.559	5.123	119.884
600.00	25.642	7.684	124.552
700.00	25.692	10.251	128.509
800.00	25.725	12.822	131.942
900.00	25.748	15.395	134.973
1000.00	25.764	17.971	137.687
1100.00	25.776	20.548	140.143
1200.00	25.785	23.126	142.386
1300.00	25.792	25.705	144.450
1400.00	25.798	28.284	146.362
1500.00	25.802	30.864	148.142
1600.00	25.806	33.445	149.807
1700.00	25.809	36.026	151.372
1800.00	25.812	38.607	152.847
1900.00	25.814	41.188	154.243
2000.00	25.816	43.769	155.567

REFERENCES

1. Dow Chemical Co. Thermal Research Laboratory. JANAF Thermochemical Tables, 2d ed., NSRDS-NBS 37, S/N 03030872, U.S. Government Printing Office, Washington, D.C., 1971, 1141 pp.

2. International Union of Pure and Applied Chemistry, Division of Physical Chemistry, Commission of Physicochemical Symbols, Terminology, and Units. Manual of Symbols and Terminology for Physicochemical Quantities and Units. 1979, p. 35.

3. King, E. G., A. D. Mah, and L. B. Pankratz. Thermodynamic Properties of Copper and Its Inorganic Compounds, INCRA Monograph Series II (sponsored by The International Copper Research Association and the U.S. Bureau of Mines), New York, 1973, pp. 5-8.

4. Mayer, J. E., and M. G. Mayer. Statistical Mechanics. John Wiley and Sons, Inc., London, 1940, 495 pp.

5. McBride, B. J., and S. Gordon. Fortran IV Program for Calculation of Thermodynamic Data. NASA TND-4097, 1967, 129 pp.

☆U.S. GOVERNMENT PRINTING OFFICE: 1981-505-002/131 INT.-BU.OF MINES,PGH.,PA. 25902